
MCSÉ-51 PROGRAMMER’S GUIDE AND INSTRUCTION SET

The information presented in this chapter is collected from the MCSÉ-51 Architectural Overview and the Hardware
Description of the 8051, 8052 and 80C51 chapters of this book. The material has been selected and rearranged to
form a quick and convenient reference for the programmers of the MCS-51. This guide pertains specifically to the
8051, 8052 and 80C51.

MEMORY ORGANIZATION

PROGRAM MEMORY

The 8051 has separate address spaces for Program Memory and Data Memory. The Program Memory can be up to
64K bytes long. The lower 4K (8K for the 8052) may reside on-chip.

Figure 1 shows a map of the 8051 program memory, and Figure 2 shows a map of the 8052 program memory.

270249–1

Figure 1. The 8051 Program Memory

270249–2

Figure 2. The 8052 Program Memory

Data Memory:

The 8051 can address up to 64K bytes of Data Memory external to the chip. The ‘‘MOVX’’ instruction is used to
access the external data memory. (Refer to the MCS-51 Instruction Set, in this chapter, for detailed description of
instructions).

The 8051 has 128 bytes of on-chip RAM (256 bytes in the 8052) plus a number of Special Function Registers (SFRs).
The lower 128 bytes of RAM can be accessed either by direct addressing (MOV data addr) or by indirect addressing
(MOV @Ri). Figure 3 shows the 8051 and the 8052 Data Memory organization.

270249–3

Figure 3a. The 8051 Data Memory

270249–4

Figure 3b. The 8052 Data Memory

INDIRECT ADDRESS AREA:

Note that in Figure 3b the SFRs and the indirect address RAM have the same addresses (80H–0FFH). Neverthe-
less, they are two separate areas and are accessed in two different ways.

For example the instruction

MOV 80H,Ý0AAH

writes 0AAH to Port 0 which is one of the SFRs and the instruction

MOV R0,Ý80H

MOV @R0,Ý0BBH

writes 0BBH in location 80H of the data RAM. Thus, after execution of both of the above instructions Port 0 will
contain 0AAH and location 80 of the RAM will contain 0BBH.

Note that the stack operations are examples of indirect addressing, so the upper 128 bytes of data RAM are available
as stack space in those devices which implement 256 bytes of internal RAM.

DIRECT AND INDIRECT ADDRESS AREA:

The 128 bytes of RAM which can be accessed by both direct and indirect addressing can be divided into 3 segments
as listed below and shown in Figure 4.

1. Register Banks 0-3: Locations 0 through 1FH (32 bytes). ASM-51 and the device after reset default to register
bank 0. To use the other register banks the user must select them in the software (refer to the MCS-51 Micro
Assembler User’s Guide). Each register bank contains 8 one-byte registers, 0 through 7.

Reset initializes the Stack Pointer to location 07H and it is incremented once to start from location 08H which is the
first register (RO) of the second register bank. Thus, in order to use more than one register bank, the SP should be
intialized to a different location of the RAM where it is not used for data storage (ie, higher part of the RAM).

2. Bit Addressable Area: 16 bytes have been assigned for this segment, 20H-2FH. Each one of the 128 bits of this
segment can be directly addressed (0-7FH).

The bits can be referred to in two ways both of which are acceptable by the ASM-51. One way is to refer to their
addresses, ie. 0 to 7FH. The other way is with reference to bytes 20H to 2FH. Thus, bits 0–7 can also be referred to
as bits 20.0–20.7, and bits 8-FH are the same as 21.0–21.7 and so on.

Each of the 16 bytes in this segment can also be addressed as a byte.

3. Scratch Pad Area: Bytes 30H through 7FH are available to the user as data RAM. However, if the stack pointer
has been initialized to this area, enough number of bytes should be left aside to prevent SP data destruction.

Figure 4 shows the different segments of the on-chip RAM.

270249–5

Figure 4. 128 Bytes of RAM Direct and Indirect Addressable

SPECIAL FUNCTION REGISTERS:

Table 1 contains a list of all the SFRs and their addresses.

Comparing Table 1 and Figure 5 shows that all of the SFRs that are byte and bit addressable are located on the first
column of the diagram in Figure 5.

Table 1

Symbol Name Address

*ACC Accumulator 0E0H

*B B Register 0F0H

*PSW Program Status Word 0D0H

SP Stack Pointer 81H

DPTR Data Pointer 2 Bytes

DPL Low Byte 82H

DPH High Byte 83H

*P0 Port 0 80H

*P1 Port 1 90H

*P2 Port 2 0A0H

*P3 Port 3 0B0H

*IP Interrupt Priority Control 0B8H

*IE Interrupt Enable Control 0A8H

TMOD Timer/Counter Mode Control 89H

*TCON Timer/Counter Control 88H

*aT2CON Timer/Counter 2 Control 0C8H

TH0 Timer/Counter 0 High Byte 8CH

TL0 Timer/Counter 0 Low Byte 8AH

TH1 Timer/Counter 1 High Byte 8DH

TL1 Timer/Counter 1 Low Byte 8BH
aTH2 Timer/Counter 2 High Byte 0CDH
aTL2 Timer/Counter 2 Low Byte 0CCH
aRCAP2H T/C 2 Capture Reg. High Byte 0CBH
aRCAP2L T/C 2 Capture Reg. Low Byte 0CAH

*SCON Serial Control 98H

SBUF Serial Data Buffer 99H

PCON Power Control 87H

* e Bit addressable
a e 8052 only

6

WHAT DO THE SFRs CONTAIN JUST AFTER POWER-ON OR A RESET?

Table 2 lists the contents of each SFR after power-on or a hardware reset.

Table 2. Contents of the SFRs after reset

Register Value in Binary

*ACC 00000000

*B 00000000

*PSW 00000000

SP 00000111

DPTR

DPH 00000000

DPL 00000000

*P0 11111111

*P1 11111111

*P2 11111111

*P3 11111111

*IP 8051 XXX00000,

8052 XX000000

*IE 8051 0XX00000,

8052 0X000000

TMOD 00000000

*TCON 00000000

*aT2CON 00000000

TH0 00000000

TL0 00000000

TH1 00000000

TL1 00000000
aTH2 00000000
aTL2 00000000
aRCAP2H 00000000
aRCAP2L 00000000

*SCON 00000000

SBUF Indeterminate

PCON HMOS 0XXXXXXX

CHMOS 0XXX0000

X e Undefined
* e Bit Addressable
a e 8052 only

SFR MEMORY MAP
8 Bytes

F8 FF

F0 B F7

E8 EF

E0 ACC E7

D8 DF

D0 PSW D7

C8 T2CON RCAP2L RCAP2H TL2 TH2 CF

C0 C7

B8 IP BF

B0 P3 B7

A8 IE AF

A0 P2 A7

98 SCON SBUF 9F

90 P1 97

88 TCON TMOD TL0 TL1 TH0 TH1 8F

80 P0 SP DPL DPH PCON 87

Figure 5u
Bit

Addressable

Those SFRs that have their bits assigned for various functions are listed in this section. A brief description of each bit
is provided for quick reference. For more detailed information refer to the Architecture Chapter of this book.

PSW: PROGRAM STATUS WORD. BIT ADDRESSABLE.

CY AC F0 RS1 RS0 OV Ð P

CY PSW.7 Carry Flag.

AC PSW.6 Auxiliary Carry Flag.

F0 PSW.5 Flag 0 available to the user for general purpose.

RS1 PSW.4 Register Bank selector bit 1 (SEE NOTE 1).

RS0 PSW.3 Register Bank selector bit 0 (SEE NOTE 1).

OV PSW.2 Overflow Flag.

Ð PSW.1 User definable flag.

P PSW.0 Parity flag. Set/cleared by hardware each instruction cycle to indicate an odd/even number of
‘1’ bits in the accumulator.

NOTE:
1. The value presented by RS0 and RS1 selects the corresponding register bank.

RS1 RS0 Register Bank Address

0 0 0 00H-07H

0 1 1 08H-0FH

1 0 2 10H-17H

1 1 3 18H-1FH

PCON: POWER CONTROL REGISTER. NOT BIT ADDRESSABLE.

SMOD Ð Ð Ð GF1 GF0 PD IDL

SMOD Double baud rate bit. If Timer 1 is used to generate baud rate and SMOD e 1, the baud rate is doubled
when the Serial Port is used in modes 1, 2, or 3.

Ð Not implemented, reserved for future use.*
Ð Not implemented, reserved for future use.*
Ð Not implemented, reserved for future use.*
GF1 General purpose flag bit.

GF0 General purpose flag bit.

PD Power Down bit. Setting this bit activates Power Down operation in the 80C51BH. (Available only in
CHMOS).

IDL Idle Mode bit. Setting this bit activates Idle Mode operation in the 80C51BH. (Available only in CHMOS).

If 1s are written to PD and IDL at the same time, PD takes precedence.

*User software should not write 1s to reserved bits. These bits may be used in future MCS-51 products to invoke new
features. In that case, the reset or inactive value of the new bit will be 0, and its active value will be 1.

9

INTERRUPTS:

In order to use any of the interrupts in the MCS-51, the following three steps must be taken.

1. Set the EA (enable all) bit in the IE register to 1.

2. Set the corresponding individual interrupt enable bit in the IE register to 1.

3. Begin the interrupt service routine at the corresponding Vector Address of that interrupt. See Table below.

Interrupt Vector

Source Address

IE0 0003H

TF0 000BH

IE1 0013H

TF1 001BH

RI & TI 0023H

TF2 & EXF2 002BH

In addition, for external interrupts, pins INT0 and INT1 (P3.2 and P3.3) must be set to 1, and depending on whether
the interrupt is to be level or transition activated, bits IT0 or IT1 in the TCON register may need to be set to 1.

ITx e 0 level activated

ITx e 1 transition activated

IE: INTERRUPT ENABLE REGISTER. BIT ADDRESSABLE.

If the bit is 0, the corresponding interrupt is disabled. If the bit is 1, the corresponding interrupt is enabled.

EA Ð ET2 ES ET1 EX1 ET0 EX0

EA IE.7 Disables all interrupts. If EA e 0, no interrupt will be acknowledged. If EA e 1, each interrupt
source is individually enabled or disabled by setting or clearing its enable bit.

Ð IE.6 Not implemented, reserved for future use.*
ET2 IE.5 Enable or disable the Timer 2 overflow or capture interrupt (8052 only).

ES IE.4 Enable or disable the serial port interrupt.

ET1 IE.3 Enable or disable the Timer 1 overflow interrupt.

EX1 IE.2 Enable or disable External Interrupt 1.

ET0 IE.1 Enable or disable the Timer 0 overflow interrupt.

EX0 IE.0 Enable or disable External Interrupt 0.

*User software should not write 1s to reserved bits. These bits may be used in future MCS-51 products to invoke
new features. In that case, the reset or inactive value of the new bit will be 0, and its active value will be 1.

10

ASSIGNING HIGHER PRIORITY TO ONE OR MORE INTERRUPTS:

In order to assign higher priority to an interrupt the corresponding bit in the IP register must be set to 1.

Remember that while an interrupt service is in progress, it cannot be interrupted by a lower or same level interrupt.

PRIORITY WITHIN LEVEL:

Priority within level is only to resolve simultaneous requests of the same priority level.

From high to low, interrupt sources are listed below:

IE0
TF0
IE1
TF1
RI or TI
TF2 or EXF2

IP: INTERRUPT PRIORITY REGISTER. BIT ADDRESSABLE.

If the bit is 0, the corresponding interrupt has a lower priority and if the bit is 1 the corresponding interrupt has a
higher priority.

Ð Ð PT2 PS PT1 PX1 PT0 PX0

Ð IP. 7 Not implemented, reserved for future use.*
Ð IP. 6 Not implemented, reserved for future use.*
PT2 IP. 5 Defines the Timer 2 interrupt priority level (8052 only).

PS IP. 4 Defines the Serial Port interrupt priority level.

PT1 IP. 3 Defines the Timer 1 interrupt priority level.

PX1 IP. 2 Defines External Interrupt 1 priority level.

PT0 IP. 1 Defines the Timer 0 interrupt priority level.

PX0 IP. 0 Defines the External Interrupt 0 priority level.

*User software should not write 1s to reserved bits. These bits may be used in future MCS-51 products to invoke
new features. In that case, the reset or inactive value of the new bit will be 0, and its active value will be 1.

11

TCON: TIMER/COUNTER CONTROL REGISTER. BIT ADDRESSABLE.

TF1 TR1 TF0 TR0 IE1 IT1 IE0 IT0

TF1 TCON. 7 Timer 1 overflow flag. Set by hardware when the Timer/Counter 1 overflows. Cleared by hard-
ware as processor vectors to the interrupt service routine.

TR1 TCON. 6 Timer 1 run control bit. Set/cleared by software to turn Timer/Counter 1 ON/OFF.

TF0 TCON. 5 Timer 0 overflow flag. Set by hardware when the Timer/Counter 0 overflows. Cleared by hard-
ware as processor vectors to the service routine.

TR0 TCON. 4 Timer 0 run control bit. Set/cleared by software to turn Timer/Counter 0 ON/OFF.

IE1 TCON. 3 External Interrupt 1 edge flag. Set by hardware when External Interrupt edge is detected.
Cleared by hardware when interrupt is processed.

IT1 TCON. 2 Interrupt 1 type control bit. Set/cleared by software to specify falling edge/low level triggered
External Interrupt.

IE0 TCON. 1 External Interrupt 0 edge flag. Set by hardware when External Interrupt edge detected. Cleared
by hardware when interrupt is processed.

IT0 TCON. 0 Interrupt 0 type control bit. Set/cleared by software to specify falling edge/low level triggered
External Interrupt.

TMOD: TIMER/COUNTER MODE CONTROL REGISTER. NOT BIT
ADDRESSABLE.

GATE C/T M1 M0 GATE C/T M1 M0X ä Y X ä Y
TIMER 1 TIMER 0

GATE When TRx (in TCON) is set and GATE e 1, TIMER/COUNTERx will run only while INTx pin is high
(hardware control). When GATE e 0, TIMER/COUNTERx will run only while TRx e 1 (software
control).

C/T Timer or Counter selector. Cleared for Timer operation (input from internal system clock). Set for Coun-
ter operation (input from Tx input pin).

M1 Mode selector bit. (NOTE 1)

M0 Mode selector bit. (NOTE 1)

NOTE 1:

M1 M0 Operating Mode

0 0 0 13-bit Timer (MCS-48 compatible)

0 1 1 16-bit Timer/Counter

1 0 2 8-bit Auto-Reload Timer/Counter

1 1 3 (Timer 0) TL0 is an 8-bit Timer/Counter controlled by the standard Timer 0
control bits, TH0 is an 8-bit Timer and is controlled by Timer 1 control bits.

1 1 3 (Timer 1) Timer/Counter 1 stopped.

12

TIMER SET-UP

Tables 3 through 6 give some values for TMOD which can be used to set up Timer 0 in different modes.

It is assumed that only one timer is being used at a time. If it is desired to run Timers 0 and 1 simultaneously, in any
mode, the value in TMOD for Timer 0 must be ORed with the value shown for Timer 1 (Tables 5 and 6).

For example, if it is desired to run Timer 0 in mode 1 GATE (external control), and Timer 1 in mode 2 COUNTER,
then the value that must be loaded into TMOD is 69H (09H from Table 3 ORed with 60H from Table 6).

Moreover, it is assumed that the user, at this point, is not ready to turn the timers on and will do that at a different
point in the program by setting bit TRx (in TCON) to 1.

TIMER/COUNTER 0

As a Timer:
Table 3

TMOD

MODE
TIMER 0 INTERNAL EXTERNAL

FUNCTION CONTROL CONTROL

(NOTE 1) (NOTE 2)

0 13-bit Timer 00H 08H

1 16-bit Timer 01H 09H

2 8-bit Auto-Reload 02H 0AH

3 two 8-bit Timers 03H 0BH

As a Counter:
Table 4

TMOD

MODE
COUNTER 0 INTERNAL EXTERNAL
FUNCTION CONTROL CONTROL

(NOTE 1) (NOTE 2)

0 13-bit Timer 04H 0CH

1 16-bit Timer 05H 0DH

2 8-bit Auto-Reload 06H 0EH

3 one 8-bit Counter 07H 0FH

NOTES:
1. The Timer is turned ON/OFF by setting/clearing bit TR0 in the software.
2. The Timer is turned ON/OFF by the 1 to 0 transition on INT0 (P3.2) when TR0 e 1
(hardware control).

13

TIMER/COUNTER 1

As a Timer:
Table 5

TMOD

MODE
TIMER 1 INTERNAL EXTERNAL

FUNCTION CONTROL CONTROL

(NOTE 1) (NOTE 2)

0 13-bit Timer 00H 80H

1 16-bit Timer 10H 90H

2 8-bit Auto-Reload 20H A0H

3 does not run 30H B0H

As a Counter:
Table 6

TMOD

MODE
COUNTER 1 INTERNAL EXTERNAL
FUNCTION CONTROL CONTROL

(NOTE 1) (NOTE 2)

0 13-bit Timer 40H C0H

1 16-bit Timer 50H D0H

2 8-bit Auto-Reload 60H E0H

3 not available Ð Ð

NOTES:
1. The Timer is turned ON/OFF by setting/clearing bit TR1 in the software.
2. The Timer is turned ON/OFF by the 1 to 0 transition on INT1 (P3.3) when TR1 e 1
(hardware control).

14

T2CON: TIMER/COUNTER 2 CONTROL REGISTER. BIT ADDRESSABLE

8052 Only

TF2 EXF2 RCLK TCLK EXEN2 TR2 C/T2 CP/RL2

TF2 T2CON. 7 Timer 2 overflow flag set by hardware and cleared by software. TF2 cannot be set when
either RCLK e 1 or CLK e 1

EXF2 T2CON. 6 Timer 2 external flag set when either a capture or reload is caused by a negative transition on
T2EX, and EXEN2 e 1. When Timer 2 interrupt is enabled, EXF2 e 1 will cause the CPU
to vector to the Timer 2 interrupt routine. EXF2 must be cleared by software.

RCLK T2CON. 5 Receive clock flag. When set, causes the Serial Port to use Timer 2 overflow pulses for its
receive clock in modes 1 & 3. RCLK e 0 causes Timer 1 overflow to be used for the receive
clock.

TLCK T2CON. 4 Transmit clock flag. When set, causes the Serial Port to use Timer 2 overflow pulses for its
transmit clock in modes 1 & 3. TCLK e 0 causes Timer 1 overflows to be used for the
transmit clock.

EXEN2 T2CON. 3 Timer 2 external enable flag. When set, allows a capture or reload to occur as a result of
negative transition on T2EX if Timer 2 is not being used to clock the Serial Port.
EXEN2 e 0 causes Timer 2 to ignore events at T2EX.

TR2 T2CON. 2 Software START/STOP control for Timer 2. A logic 1 starts the Timer.

C/T2 T2CON. 1 Timer or Counter select.

0 e Internal Timer. 1 e External Event Counter (falling edge triggered).

CP/RL2 T2CON. 0 Capture/Reload flag. When set, captures will occur on negative transitions at T2EX if
EXEN2 e 1. When cleared, Auto-Reloads will occur either with Timer 2 overflows or
negative transitions at T2EX when EXEN2 e 1. When either RCLK e 1 or TCLK e 1,
this bit is ignored and the Timer is forced to Auto-Reload on Timer 2 overflow.

15

TIMER/COUNTER 2 SET-UP

Except for the baud rate generator mode, the values given for T2CON do not include the setting of the TR2 bit.
Therefore, bit TR2 must be set, separately, to turn the Timer on.

As a Timer:
Table 7

T2CON

MODE INTERNAL EXTERNAL

CONTROL CONTROL

(NOTE 1) (NOTE 2)

16-bit Auto-Reload 00H 08H

16-bit Capture 01H 09H

BAUD rate generator receive &

transmit same baud rate 34H 36H

receive only 24H 26H

transmit only 14H 16H

As a Counter:
Table 8

TMOD

MODE INTERNAL EXTERNAL

CONTROL CONTROL

(NOTE 1) (NOTE 2)

16-bit Auto-Reload 02H 0AH

16-bit Capture 03H 0BH

NOTES:
1. Capture/Reload occurs only on Timer/Counter overflow.
2. Capture/Reload occurs on Timer/Counter overflow and a 1 to 0 transition on T2EX
(P1.1) pin except when Timer 2 is used in the baud rate generating mode.

16

SCON: SERIAL PORT CONTROL REGISTER. BIT ADDRESSABLE.

SM0 SM1 SM2 REN TB8 RB8 TI RI

SM0 SCON. 7 Serial Port mode specifier. (NOTE 1).

SM1 SCON. 6 Serial Port mode specifier. (NOTE 1).

SM2 SCON. 5 Enables the multiprocessor communication feature in modes 2 & 3. In mode 2 or 3, if SM2 is set
to 1 then RI will not be activated if the received 9th data bit (RB8) is 0. In mode 1, if SM2 e 1
then RI will not be activated if a valid stop bit was not received. In mode 0, SM2 should be 0.
(See Table 9).

REN SCON. 4 Set/Cleared by software to Enable/Disable reception.

TB8 SCON. 3 The 9th bit that will be transmitted in modes 2 & 3. Set/Cleared by software.

RB8 SCON. 2 In modes 2 & 3, is the 9th data bit that was received. In mode 1, if SM2 e 0, RB8 is the stop bit
that was received. In mode 0, RB8 is not used.

TI SCON. 1 Transmit interrupt flag. Set by hardware at the end of the 8th bit time in mode 0, or at the
beginning of the stop bit in the other modes. Must be cleared by software.

RI SCON. 0 Receive interrupt flag. Set by hardware at the end of the 8th bit time in mode 0, or halfway
through the stop bit time in the other modes (except see SM2). Must be cleared by software.

NOTE 1:

SM0 SM1 Mode Description Baud Rate

0 0 0 SHIFT REGISTER Fosc./12

0 1 1 8-Bit UART Variable

1 0 2 9-Bit UART Fosc./64 OR

Fosc./32

1 1 3 9-Bit UART Variable

SERIAL PORT SET-UP:
Table 9

MODE SCON SM2 VARIATION

0 10H
Single Processor

1 50H
Environment

2 90H
(SM2 e 0)

3 D0H

0 NA
Multiprocessor

1 70H
Environment

2 B0H
(SM2 e 1)

3 F0H

GENERATING BAUD RATES

Serial Port in Mode 0:

Mode 0 has a fixed baud rate which is 1/12 of the oscillator frequency. To run the serial port in this mode none of
the Timer/Counters need to be set up. Only the SCON register needs to be defined.

Baud Rate e

Osc Freq

12

Serial Port in Mode 1:

Mode 1 has a variable baud rate. The baud rate can be generated by either Timer 1 or Timer 2 (8052 only).

17

USING TIMER/COUNTER 1 TO GENERATE BAUD RATES:

For this purpose, Timer 1 is used in mode 2 (Auto-Reload). Refer to Timer Setup section of this chapter.

Baud Rate e

K x Oscillator Freq.

32 x 12 x [256 b (TH1)]

If SMOD e 0, then K e 1.
If SMOD e 1, then K e 2. (SMOD is the PCON register).

Most of the time the user knows the baud rate and needs to know the reload value for TH1.
Therefore, the equation to calculate TH1 can be written as:

TH1 e 256 b

K x Osc Freq.

384 x baud rate

TH1 must be an integer value. Rounding off TH1 to the nearest integer may not produce the desired baud rate. In
this case, the user may have to choose another crystal frequency.

Since the PCON register is not bit addressable, one way to set the bit is logical ORing the PCON register. (ie, ORL
PCON,Ý80H). The address of PCON is 87H.

USING TIMER/COUNTER 2 TO GENERATE BAUD RATES:

For this purpose, Timer 2 must be used in the baud rate generating mode. Refer to Timer 2 Setup Table in this
chapter. If Timer 2 is being clocked through pin T2 (P1.0) the baud rate is:

Baud Rate e

Timer 2 Overflow Rate

16

And if it is being clocked internally the baud rate is:

Baud Rate e

Osc Freq

32 x [65536 b (RCAP2H, RCAP2L)]

To obtain the reload value for RCAP2H and RCAP2L the above equation can be rewritten as:

RCAP2H, RCAP2L e 65536 b

Osc Freq

32 x Baud Rate

SERIAL PORT IN MODE 2:

The baud rate is fixed in this mode and is (/32 or (/64 of the oscillator frequency depending on the value of the SMOD
bit in the PCON register.

In this mode none of the Timers are used and the clock comes from the internal phase 2 clock.

SMOD e 1, Baud Rate e (/32 Osc Freq.

SMOD e 0, Baud Rate e (/64 Osc Freq.

To set the SMOD bit: ORL PCON,Ý80H. The address of PCON is 87H.

SERIAL PORT IN MODE 3:

The baud rate in mode 3 is variable and sets up exactly the same as in mode 1.

18

MCSÉ-51 INSTRUCTION SET

Table 10. 8051 Instruction Set Summary

Interrupt Response Time: Refer to Hardware De-
scription Chapter.

Instructions that Affect Flag Settings(1)

Instruction Flag Instruction Flag

C OV AC C OV AC

ADD X X X CLR C O

ADDC X X X CPL C X

SUBB X X X ANL C,bit X

MUL O X ANL C,/bit X

DIV O X ORL C,bit X

DA X ORL C,bit X

RRC X MOV C,bit X

RLC X CJNE X

SETB C 1

(1)Note that operations on SFR byte address 208 or
bit addresses 209-215 (i.e., the PSW or bits in the
PSW) will also affect flag settings.

Note on instruction set and addressing modes:

Rn Ð Register R7–R0 of the currently se-
lected Register Bank.

direct Ð 8-bit internal data location’s address.
This could be an Internal Data RAM
location (0–127) or a SFR [i.e., I/O
port, control register, status register,
etc. (128–255)].

@Ri Ð 8-bit internal data RAM location (0–
255) addressed indirectly through reg-
ister R1 or R0.

Ýdata Ð 8-bit constant included in instruction.
Ýdata 16 Ð 16-bit constant included in instruction.
addr 16 Ð 16-bit destination address. Used by

LCALL & LJMP. A branch can be
anywhere within the 64K-byte Pro-
gram Memory address space.

addr 11 Ð 11-bit destination address. Used by
ACALL & AJMP. The branch will be
within the same 2K-byte page of pro-
gram memory as the first byte of the
following instruction.

rel Ð Signed (two’s complement) 8-bit offset
byte. Used by SJMP and all condition-
al jumps. Range is b128 to a127
bytes relative to first byte of the fol-
lowing instruction.

bit Ð Direct Addressed bit in Internal Data
RAM or Special Function Register.

Mnemonic Description Byte
Oscillator

Period

ARITHMETIC OPERATIONS

ADD A,Rn Add register to 1 12

Accumulator

ADD A,direct Add direct byte to 2 12

Accumulator

ADD A,@Ri Add indirect RAM 1 12

to Accumulator

ADD A,Ýdata Add immediate 2 12

data to

Accumulator

ADDC A,Rn Add register to 1 12

Accumulator

with Carry

ADDC A,direct Add direct byte to 2 12

Accumulator

with Carry

ADDC A,@Ri Add indirect 1 12

RAM to

Accumulator

with Carry

ADDC A,Ýdata Add immediate 2 12

data to Acc

with Carry

SUBB A,Rn Subtract Register 1 12

from Acc with

borrow

SUBB A,direct Subtract direct 2 12

byte from Acc

with borrow

SUBB A,@Ri Subtract indirect 1 12

RAM from ACC

with borrow

SUBB A,Ýdata Subtract 2 12

immediate data

from Acc with

borrow

INC A Increment 1 12

Accumulator

INC Rn Increment register 1 12

INC direct Increment direct 2 12

byte

INC @Ri Increment direct 1 12

RAM

DEC A Decrement 1 12

Accumulator

DEC Rn Decrement 1 12

Register

DEC direct Decrement direct 2 12

byte

DEC @Ri Decrement 1 12

indirect RAM

All mnemonics copyrighted ©Intel Corporation 1980

Table 10. 8051 Instruction Set Summary (Continued)

Mnemonic Description Byte
Oscillator

Period

ARITHMETIC OPERATIONS (Continued)

INC DPTR Increment Data 1 24

Pointer

MUL AB Multiply A & B 1 48

DIV AB Divide A by B 1 48

DA A Decimal Adjust 1 12

Accumulator

LOGICAL OPERATIONS

ANL A,Rn AND Register to 1 12

Accumulator

ANL A,direct AND direct byte 2 12

to Accumulator

ANL A,@Ri AND indirect 1 12

RAM to

Accumulator

ANL A,Ýdata AND immediate 2 12

data to

Accumulator

ANL direct,A AND Accumulator 2 12

to direct byte

ANL direct,Ýdata AND immediate 3 24

data to direct byte

ORL A,Rn OR register to 1 12

Accumulator

ORL A,direct OR direct byte to 2 12

Accumulator

ORL A,@Ri OR indirect RAM 1 12

to Accumulator

ORL A,Ýdata OR immediate 2 12

data to

Accumulator

ORL direct,A OR Accumulator 2 12

to direct byte

ORL direct,Ýdata OR immediate 3 24

data to direct byte

XRL A,Rn Exclusive-OR 1 12

register to

Accumulator

XRL A,direct Exclusive-OR 2 12

direct byte to

Accumulator

XRL A,@Ri Exclusive-OR 1 12

indirect RAM to

Accumulator

XRL A,Ýdata Exclusive-OR 2 12

immediate data to

Accumulator

XRL direct,A Exclusive-OR 2 12

Accumulator to

direct byte

XRL direct,Ýdata Exclusive-OR 3 24

immediate data

to direct byte

CLR A Clear 1 12

Accumulator

CPL A Complement 1 12

Accumulator

Mnemonic Description Byte
Oscillator

Period

LOGICAL OPERATIONS (Continued)

RL A Rotate 1 12

Accumulator Left

RLC A Rotate 1 12

Accumulator Left

through the Carry

RR A Rotate 1 12

Accumulator

Right

RRC A Rotate 1 12

Accumulator

Right through

the Carry

SWAP A Swap nibbles 1 12

within the

Accumulator

DATA TRANSFER

MOV A,Rn Move 1 12

register to

Accumulator

MOV A,direct Move direct 2 12

byte to

Accumulator

MOV A,@Ri Move indirect 1 12

RAM to

Accumulator

MOV A,Ýdata Move 2 12

immediate

data to

Accumulator

MOV Rn,A Move 1 12

Accumulator

to register

MOV Rn,direct Move direct 2 24

byte to

register

MOV Rn,Ýdata Move 2 12

immediate data

to register

MOV direct,A Move 2 12

Accumulator

to direct byte

MOV direct,Rn Move register 2 24

to direct byte

MOV direct,direct Move direct 3 24

byte to direct

MOV direct,@Ri Move indirect 2 24

RAM to

direct byte

MOV direct,Ýdata Move 3 24

immediate data

to direct byte

MOV @Ri,A Move 1 12

Accumulator to

indirect RAM

All mnemonics copyrighted ©Intel Corporation 1980

Table 10. 8051 Instruction Set Summary (Continued)

Mnemonic Description Byte
Oscillator

Period

DATA TRANSFER (Continued)

MOV @Ri,direct Move direct 2 24

byte to

indirect RAM

MOV @Ri,Ýdata Move 2 12

immediate

data to

indirect RAM

MOV DPTR,Ýdata16 Load Data 3 24

Pointer with a

16-bit constant

MOVC A,@AaDPTR Move Code 1 24

byte relative to

DPTR to Acc

MOVC A,@AaPC Move Code 1 24

byte relative to

PC to Acc

MOVX A,@Ri Move 1 24

External

RAM (8-bit

addr) to Acc

MOVX A,@DPTR Move 1 24

External

RAM (16-bit

addr) to Acc

MOVX @Ri,A Move Acc to 1 24

External RAM

(8-bit addr)

MOVX @DPTR,A Move Acc to 1 24

External RAM

(16-bit addr)

PUSH direct Push direct 2 24

byte onto

stack

POP direct Pop direct 2 24

byte from

stack

XCH A,Rn Exchange 1 12

register with

Accumulator

XCH A,direct Exchange 2 12

direct byte

with

Accumulator

XCH A,@Ri Exchange 1 12

indirect RAM

with

Accumulator

XCHD A,@Ri Exchange low- 1 12

order Digit

indirect RAM

with Acc

Mnemonic Description Byte
Oscillator

Period

BOOLEAN VARIABLE MANIPULATION

CLR C Clear Carry 1 12

CLR bit Clear direct bit 2 12

SETB C Set Carry 1 12

SETB bit Set direct bit 2 12

CPL C Complement 1 12

Carry

CPL bit Complement 2 12

direct bit

ANL C,bit AND direct bit 2 24

to CARRY

ANL C,/bit AND complement 2 24

of direct bit

to Carry

ORL C,bit OR direct bit 2 24

to Carry

ORL C,/bit OR complement 2 24

of direct bit

to Carry

MOV C,bit Move direct bit 2 12

to Carry

MOV bit,C Move Carry to 2 24

direct bit

JC rel Jump if Carry 2 24

is set

JNC rel Jump if Carry 2 24

not set

JB bit,rel Jump if direct 3 24

Bit is set

JNB bit,rel Jump if direct 3 24

Bit is Not set

JBC bit,rel Jump if direct 3 24

Bit is set &

clear bit

PROGRAM BRANCHING

ACALL addr11 Absolute 2 24

Subroutine

Call

LCALL addr16 Long 3 24

Subroutine

Call

RET Return from 1 24

Subroutine

RETI Return from 1 24

interrupt

AJMP addr11 Absolute 2 24

Jump

LJMP addr16 Long Jump 3 24

SJMP rel Short Jump 2 24

(relative addr)

All mnemonics copyrighted ©Intel Corporation 1980

Table 10. 8051 Instruction Set Summary (Continued)

Mnemonic Description Byte
Oscillator

Period

PROGRAM BRANCHING (Continued)

JMP @AaDPTR Jump indirect 1 24

relative to the

DPTR

JZ rel Jump if 2 24

Accumulator

is Zero

JNZ rel Jump if 2 24

Accumulator

is Not Zero

CJNE A,direct,rel Compare 3 24

direct byte to

Acc and Jump

if Not Equal

CJNE A,Ýdata,rel Compare 3 24

immediate to

Acc and Jump

if Not Equal

Mnemonic Description Byte
Oscillator

Period

PROGRAM BRANCHING (Continued)

CJNE Rn,Ýdata,rel Compare 3 24

immediate to

register and

Jump if Not

Equal

CJNE @Ri,Ýdata,rel Compare 3 24

immediate to

indirect and

Jump if Not

Equal

DJNZ Rn,rel Decrement 2 24

register and

Jump if Not

Zero

DJNZ direct,rel Decrement 3 24

direct byte

and Jump if

Not Zero

NOP No Operation 1 12

All mnemonics copyrighted ©Intel Corporation 1980

Table 11. Instruction Opcodes in Hexadecimal Order

Hex Number
Mnemonic Operands

Code of Bytes

00 1 NOP

01 2 AJMP code addr

02 3 LJMP code addr

03 1 RR A

04 1 INC A

05 2 INC data addr

06 1 INC @R0

07 1 INC @R1

08 1 INC R0

09 1 INC R1

0A 1 INC R2

0B 1 INC R3

0C 1 INC R4

0D 1 INC R5

0E 1 INC R6

0F 1 INC R7

10 3 JBC bit addr, code addr

11 2 ACALL code addr

12 3 LCALL code addr

13 1 RRC A

14 1 DEC A

15 2 DEC data addr

16 1 DEC @R0

17 1 DEC @R1

18 1 DEC R0

19 1 DEC R1

1A 1 DEC R2

1B 1 DEC R3

1C 1 DEC R4

1D 1 DEC R5

1E 1 DEC R6

1F 1 DEC R7

20 3 JB bit addr, code addr

21 2 AJMP code addr

22 1 RET

23 1 RL A

24 2 ADD A,Ýdata

25 2 ADD A,data addr

26 1 ADD A,@R0

27 1 ADD A,@R1

28 1 ADD A,R0

29 1 ADD A,R1

2A 1 ADD A,R2

2B 1 ADD A,R3

2C 1 ADD A,R4

2D 1 ADD A,R5

2E 1 ADD A,R6

2F 1 ADD A,R7

30 3 JNB bit addr, code addr

31 2 ACALL code addr

32 1 RETI

Hex Number
Mnemonic Operands

Code of Bytes

33 1 RLC A

34 2 ADDC A,Ýdata

35 2 ADDC A,data addr

36 1 ADDC A,@R0

37 1 ADDC A,@R1

38 1 ADDC A,R0

39 1 ADDC A,R1

3A 1 ADDC A,R2

3B 1 ADDC A,R3

3C 1 ADDC A,R4

3D 1 ADDC A,R5

3E 1 ADDC A,R6

3F 1 ADDC A,R7

40 2 JC code addr

41 2 AJMP code addr

42 2 ORL data addr,A

43 3 ORL data addr,Ýdata

44 2 ORL A,Ýdata

45 2 ORL A,data addr

46 1 ORL A,@R0

47 1 ORL A,@R1

48 1 ORL A,R0

49 1 ORL A,R1

4A 1 ORL A,R2

4B 1 ORL A,R3

4C 1 ORL A,R4

4D 1 ORL A,R5

4E 1 ORL A,R6

4F 1 ORL A,R7

50 2 JNC code addr

51 2 ACALL code addr

52 2 ANL data addr,A

53 3 ANL data addr,Ýdata

54 2 ANL A,Ýdata

55 2 ANL A,data addr

56 1 ANL A,@R0

57 1 ANL A,@R1

58 1 ANL A,R0

59 1 ANL A,R1

5A 1 ANL A,R2

5B 1 ANL A,R3

5C 1 ANL A,R4

5D 1 ANL A,R5

5E 1 ANL A,R6

5F 1 ANL A,R7

60 2 JZ code addr

61 2 AJMP code addr

62 2 XRL data addr,A

63 3 XRL data addr,Ýdata

64 2 XRL A,Ýdata

65 2 XRL A,data addr

Table 11. Instruction Opcodes in Hexadecimal Order (Continued)

Hex Number
Mnemonic Operands

Code of Bytes

66 1 XRL A,@R0

67 1 XRL A,@R1

68 1 XRL A,R0

69 1 XRL A,R1

6A 1 XRL A,R2

6B 1 XRL A,R3

6C 1 XRL A,R4

6D 1 XRL A,R5

6E 1 XRL A,R6

6F 1 XRL A,R7

70 2 JNZ code addr

71 2 ACALL code addr

72 2 ORL C,bit addr

73 1 JMP @AaDPTR

74 2 MOV A,Ýdata

75 3 MOV data addr,Ýdata

76 2 MOV @R0,Ýdata

77 2 MOV @R1,Ýdata

78 2 MOV R0,Ýdata

79 2 MOV R1,Ýdata

7A 2 MOV R2,Ýdata

7B 2 MOV R3,Ýdata

7C 2 MOV R4,Ýdata

7D 2 MOV R5,Ýdata

7E 2 MOV R6,Ýdata

7F 2 MOV R7,Ýdata

80 2 SJMP code addr

81 2 AJMP code addr

82 2 ANL C,bit addr

83 1 MOVC A,@AaPC

84 1 DIV AB

85 3 MOV data addr, data addr

86 2 MOV data addr,@R0

87 2 MOV data addr,@R1

88 2 MOV data addr,R0

89 2 MOV data addr,R1

8A 2 MOV data addr,R2

8B 2 MOV data addr,R3

8C 2 MOV data addr,R4

8D 2 MOV data addr,R5

8E 2 MOV data addr,R6

8F 2 MOV data addr,R7

90 3 MOV DPTR,Ýdata

91 2 ACALL code addr

92 2 MOV bit addr,C

93 1 MOVC A,@AaDPTR

94 2 SUBB A,Ýdata

95 2 SUBB A,data addr

96 1 SUBB A,@R0

97 1 SUBB A,@R1

98 1 SUBB A,R0

Hex Number
Mnemonic Operands

Code of Bytes

99 1 SUBB A,R1

9A 1 SUBB A,R2

9B 1 SUBB A,R3

9C 1 SUBB A,R4

9D 1 SUBB A,R5

9E 1 SUBB A,R6

9F 1 SUBB A,R7

A0 2 ORL C,/bit addr

A1 2 AJMP code addr

A2 2 MOV C,bit addr

A3 1 INC DPTR

A4 1 MUL AB

A5 reserved

A6 2 MOV @R0,data addr

A7 2 MOV @R1,data addr

A8 2 MOV R0,data addr

A9 2 MOV R1,data addr

AA 2 MOV R2,data addr

AB 2 MOV R3,data addr

AC 2 MOV R4,data addr

AD 2 MOV R5,data addr

AE 2 MOV R6,data addr

AF 2 MOV R7,data addr

B0 2 ANL C,/bit addr

B1 2 ACALL code addr

B2 2 CPL bit addr

B3 1 CPL C

B4 3 CJNE A,Ýdata,code addr

B5 3 CJNE A,data addr,code addr

B6 3 CJNE @R0,Ýdata,code addr

B7 3 CJNE @R1,Ýdata,code addr

B8 3 CJNE R0,Ýdata,code addr

B9 3 CJNE R1,Ýdata,code addr

BA 3 CJNE R2,Ýdata,code addr

BB 3 CJNE R3,Ýdata,code addr

BC 3 CJNE R4,Ýdata,code addr

BD 3 CJNE R5,Ýdata,code addr

BE 3 CJNE R6,Ýdata,code addr

BF 3 CJNE R7,Ýdata,code addr

C0 2 PUSH data addr

C1 2 AJMP code addr

C2 2 CLR bit addr

C3 1 CLR C

C4 1 SWAP A

C5 2 XCH A,data addr

C6 1 XCH A,@R0

C7 1 XCH A,@R1

C8 1 XCH A,R0

C9 1 XCH A,R1

CA 1 XCH A,R2

CB 1 XCH A,R3

Table 11. Instruction Opcodes in Hexadecimal Order (Continued)

Hex Number
Mnemonic Operands

Code of Bytes

CC 1 XCH A,R4

CD 1 XCH A,R5

CE 1 XCH A,R6

CF 1 XCH A,R7

D0 2 POP data addr

D1 2 ACALL code addr

D2 2 SETB bit addr

D3 1 SETB C

D4 1 DA A

D5 3 DJNZ data addr,code addr

D6 1 XCHD A,@R0

D7 1 XCHD A,@R1

D8 2 DJNZ R0,code addr

D9 2 DJNZ R1,code addr

DA 2 DJNZ R2,code addr

DB 2 DJNZ R3,code addr

DC 2 DJNZ R4,code addr

DD 2 DJNZ R5,code addr

DE 2 DJNZ R6,code addr

DF 2 DJNZ R7,code addr

E0 1 MOVX A,@DPTR

E1 2 AJMP code addr

E2 1 MOVX A,@R0

E3 1 MOVX A,@R1

E4 1 CLR A

E5 2 MOV A,data addr

Hex Number
Mnemonic Operands

Code of Bytes

E6 1 MOV A,@R0

E7 1 MOV A,@R1

E8 1 MOV A,R0

E9 1 MOV A,R1

EA 1 MOV A,R2

EB 1 MOV A,R3

EC 1 MOV A,R4

ED 1 MOV A,R5

EE 1 MOV A,R6

EF 1 MOV A,R7

F0 1 MOVX @DPTR,A

F1 2 ACALL code addr

F2 1 MOVX @R0,A

F3 1 MOVX @R1,A

F4 1 CPL A

F5 2 MOV data addr,A

F6 1 MOV @R0,A

F7 1 MOV @R1,A

F8 1 MOV R0,A

F9 1 MOV R1,A

FA 1 MOV R2,A

FB 1 MOV R3,A

FC 1 MOV R4,A

FD 1 MOV R5,A

FE 1 MOV R6,A

FF 1 MOV R7,A

INSTRUCTION DEFINITIONS

ACALL addr11

Function: Absolute Call

Description: ACALL unconditionally calls a subroutine located at the indicated address. The instruction
increments the PC twice to obtain the address of the following instruction, then pushes the
16-bit result onto the stack (low-order byte first) and increments the Stack Pointer twice. The
destination address is obtained by successively concatenating the five high-order bits of the
incremented PC, opcode bits 7-5, and the second byte of the instruction. The subroutine called
must therefore start within the same 2K block of the program memory as the first byte of the
instruction following ACALL. No flags are affected.

Example: Initially SP equals 07H. The label ‘‘SUBRTN’’ is at program memory location 0345 H. After
executing the instruction,

ACALL SUBRTN

at location 0123H, SP will contain 09H, internal RAM locations 08H and 09H will contain
25H and 01H, respectively, and the PC will contain 0345H.

Bytes: 2

Cycles: 2

Encoding: a10 a9 a8 1 0 0 0 1 a7 a6 a5 a4 a3 a2 a1 a0

Operation: ACALL
(PC)w(PC) a 2
(SP)w(SP) a 1
((SP))w(PC7-0)
(SP)w(SP) a 1
((SP))w(PC15-8)
(PC10-0)wpage address

ADD A,ksrc-bytel

Function: Add

Description: ADD adds the byte variable indicated to the Accumulator, leaving the result in the Accumula-
tor. The carry and auxiliary-carry flags are set, respectively, if there is a carry-out from bit 7 or
bit 3, and cleared otherwise. When adding unsigned integers, the carry flag indicates an
overflow occured.

OV is set if there is a carry-out of bit 6 but not out of bit 7, or a carry-out of bit 7 but not bit 6;
otherwise OV is cleared. When adding signed integers, OV indicates a negative number pro-
duced as the sum of two positive operands, or a positive sum from two negative operands.

Four source operand addressing modes are allowed: register, direct, register-indirect, or imme-
diate.

Example: The Accumulator holds 0C3H (11000011B) and register 0 holds 0AAH (10101010B). The
instruction,

ADD A,R0

will leave 6DH (01101101B) in the Accumulator with the AC flag cleared and both the carry
flag and OV set to 1.

ADD A,Rn

Bytes: 1

Cycles: 1

Encoding: 0 0 1 0 1 r r r

Operation: ADD
(A)w(A) a (Rn)

ADD A,direct

Bytes: 2

Cycles: 1

Encoding: 0 0 1 0 0 1 0 1 direct address

Operation: ADD
(A)w(A) a (direct)

ADD A,@Ri

Bytes: 1

Cycles: 1

Encoding: 0 0 1 0 0 1 1 i

Operation: ADD
(A)w(A) a ((Ri))

ADD A,Ýdata

Bytes: 2

Cycles: 1

Encoding: 0 0 1 0 0 1 0 0 immediate data

Operation: ADD
(A)w(A) a Ýdata

ADDC A,ksrc-bytel

Function: Add with Carry

Description: ADDC simultaneously adds the byte variable indicated, the carry flag and the Accumulator
contents, leaving the result in the Accumulator. The carry and auxiliary-carry flags are set,
respectively, if there is a carry-out from bit 7 or bit 3, and cleared otherwise. When adding
unsigned integers, the carry flag indicates an overflow occured.

OV is set if there is a carry-out of bit 6 but not out of bit 7, or a carry-out of bit 7 but not out of
bit 6; otherwise OV is cleared. When adding signed integers, OV indicates a negative number
produced as the sum of two positive operands or a positive sum from two negative operands.

Four source operand addressing modes are allowed: register, direct, register-indirect, or imme-
diate.

Example: The Accumulator holds 0C3H (11000011B) and register 0 holds 0AAH (10101010B) with the
carry flag set. The instruction,

ADDC A,R0

will leave 6EH (01101110B) in the Accumulator with AC cleared and both the Carry flag and
OV set to 1.

ADDC A,Rn

Bytes: 1

Cycles: 1

Encoding: 0 0 1 1 1 r r r

Operation: ADDC
(A)w(A) a (C) a(Rn)

ADDC A,direct

Bytes: 2

Cycles: 1

Encoding: 0 0 1 1 0 1 0 1 direct address

Operation: ADDC
(A)w(A) a (C) a (direct)

ADDC A,@Ri

Bytes: 1

Cycles: 1

Encoding: 0 0 1 1 0 1 1 i

Operation: ADDC
(A)w(A) a (C) a ((Ri))

ADDC A,Ýdata

Bytes: 2

Cycles: 1

Encoding: 0 0 1 1 0 1 0 0 immediate data

Operation: ADDC
(A)w(A) a (C) a Ýdata

AJMP addr11

Function: Absolute Jump

Description: AJMP transfers program execution to the indicated address, which is formed at run-time by
concatenating the high-order five bits of the PC (after incrementing the PC twice), opcode bits
7-5, and the second byte of the instruction. The destination must therefore be within the same
2K block of program memory as the first byte of the instruction following AJMP.

Example: The label ‘‘JMPADR’’ is at program memory location 0123H. The instruction,

AJMP JMPADR

is at location 0345H and will load the PC with 0123H.

Bytes: 2

Cycles: 2

Encoding: a10 a9 a8 0 0 0 0 1 a7 a6 a5 a4 a3 a2 a1 a0

Operation: AJMP
(PC)w(PC) a 2
(PC10-0)wpage address

ANL kdest-bytel,ksrc-bytel

Function: Logical-AND for byte variables

Description: ANL performs the bitwise logical-AND operation between the variables indicated and stores
the results in the destination variable. No flags are affected.

The two operands allow six addressing mode combinations. When the destination is the Accu-
mulator, the source can use register, direct, register-indirect, or immediate addressing; when
the destination is a direct address, the source can be the Accumulator or immediate data.

Note: When this instruction is used to modify an output port, the value used as the original
port data will be read from the output data latch, not the input pins.

Example: If the Accumulator holds 0C3H (11000011B) and register 0 holds 55H (01010101B) then the
instruction,

ANL A,R0

will leave 41H (01000001B) in the Accumulator.

When the destination is a directly addressed byte, this instruction will clear combinations of
bits in any RAM location or hardware register. The mask byte determining the pattern of bits
to be cleared would either be a constant contained in the instruction or a value computed in
the Accumulator at run-time. The instruction,

ANL P1,Ý01110011B

will clear bits 7, 3, and 2 of output port 1.

ANL A,Rn

Bytes: 1

Cycles: 1

Encoding: 0 1 0 1 1 r r r

Operation: ANL
(A)w(A) ! (Rn)

ANL A,direct

Bytes: 2

Cycles: 1

Encoding: 0 1 0 1 0 1 0 1 direct address

Operation: ANL

(A)w(A) ! (direct)

ANL A,@Ri

Bytes: 1

Cycles: 1

Encoding: 0 1 0 1 0 1 1 i

Operation: ANL

(A)w(A) ! ((Ri))

ANL A,Ýdata

Bytes: 2

Cycles: 1

Encoding: 0 1 0 1 0 1 0 0 immediate data

Operation: ANL
(A)w(A) ! Ýdata

ANL direct,A

Bytes: 2

Cycles: 1

Encoding: 0 1 0 1 0 0 1 0 direct address

Operation: ANL
(direct)w(direct) ! (A)

ANL direct,Ýdata

Bytes: 3

Cycles: 2

Encoding: 0 1 0 1 0 0 1 1 direct address immediate data

Operation: ANL
(direct)w(direct) ! Ýdata

ANL C,ksrc-bitl

Function: Logical-AND for bit variables

Description: If the Boolean value of the source bit is a logical 0 then clear the carry flag; otherwise leave the
carry flag in its current state. A slash (‘‘/’’) preceding the operand in the assembly language
indicates that the logical complement of the addressed bit is used as the source value, but the
source bit itself is not affected. No other flags are affected.

Only direct addressing is allowed for the source operand.
Example: Set the carry flag if, and only if, P1.0 e 1, ACC. 7 e 1, and OV e 0:

MOV C,P1.0 ;LOAD CARRY WITH INPUT PIN STATE

ANL C,ACC.7 ;AND CARRY WITH ACCUM. BIT 7

ANL C,/OV ;AND WITH INVERSE OF OVERFLOW FLAG

ANL C,bit

Bytes: 2

Cycles: 2

Encoding: 1 0 0 0 0 0 1 0 bit address

Operation: ANL
(C)w(C) ! (bit)

ANL C,/bit

Bytes: 2

Cycles: 2

Encoding: 1 0 1 1 0 0 0 0 bit address

Operation: ANL
(C)w(C) ! s (bit)

CJNE kdest-bytel,ksrc-bytel, rel

Function: Compare and Jump if Not Equal.

Description: CJNE compares the magnitudes of the first two operands, and branches if their values are not
equal. The branch destination is computed by adding the signed relative-displacement in the
last instruction byte to the PC, after incrementing the PC to the start of the next instruction.
The carry flag is set if the unsigned integer value of kdest-bytel is less than the unsigned
integer value of ksrc-bytel; otherwise, the carry is cleared. Neither operand is affected.

The first two operands allow four addressing mode combinations: the Accumulator may be
compared with any directly addressed byte or immediate data, and any indirect RAM location
or working register can be compared with an immediate constant.

Example: The Accumulator contains 34H. Register 7 contains 56H. The first instruction in the se-
quence,

CJNE R7,Ý60H, NOTÐEQ
; . . . . . . . . ; R7 e 60H.
NOTÐEQ: JC REQÐLOW ; IF R7 k 60H.
; . . . . . . . . ; R7 l 60H.

sets the carry flag and branches to the instruction at label NOTÐEQ. By testing the carry flag,
this instruction determines whether R7 is greater or less than 60H.

If the data being presented to Port 1 is also 34H, then the instruction,

WAIT: CJNE A,P1,WAIT

clears the carry flag and continues with the next instruction in sequence, since the Accumula-
tor does equal the data read from P1. (If some other value was being input on P1, the program
will loop at this point until the P1 data changes to 34H.)

CJNE A,direct,rel

Bytes: 3

Cycles: 2

Encoding: 1 0 1 1 0 1 0 1 direct address rel. address

Operation: (PC)w(PC) a 3
IF (A) kl (direct)
THEN

(PC)w(PC) a relative offset

IF (A) k (direct)
THEN

(C)w1
ELSE

(C)w0

CJNE A,Ýdata,rel

Bytes: 3

Cycles: 2

Encoding: 1 0 1 1 0 1 0 0 immediate data rel. address

Operation: (PC)w(PC) a 3
IF (A) kl data
THEN

(PC)w(PC) a relative offset

IF (A) k data
THEN

(C)w1
ELSE

(C)w0

CJNE Rn,Ýdata,rel

Bytes: 3

Cycles: 2

Encoding: 1 0 1 1 1 r r r immediate data rel. address

Operation: (PC)w(PC) a 3
IF (Rn) kl data
THEN

(PC)w(PC) a relative offset

IF (Rn) k data
THEN

(C)w1
ELSE

(C)w0

CJNE @Ri,Ýdata,rel

Bytes: 3

Cycles: 2

Encoding: 1 0 1 1 0 1 1 i immediate data rel. address

Operation: (PC)w(PC) a 3
IF ((Ri)) kl data
THEN

(PC)w(PC) a relative offset

IF ((Ri)) k data
THEN

(C)w1
ELSE

(C)w0

CLR A

Function: Clear Accumulator

Description: The Accumulator is cleared (all bits set on zero). No flags are affected.

Example: The Accumulator contains 5CH (01011100B). The instruction,

CLR A

will leave the Accumulator set to 00H (00000000B).

Bytes: 1

Cycles: 1

Encoding: 1 1 1 0 0 1 0 0

Operation: CLR
(A)w0

CLR bit

Function: Clear bit

Description: The indicated bit is cleared (reset to zero). No other flags are affected. CLR can operate on the
carry flag or any directly addressable bit.

Example: Port 1 has previously been written with 5DH (01011101B). The instruction,

CLR P1.2

will leave the port set to 59H (01011001B).

CLR C

Bytes: 1

Cycles: 1

Encoding: 1 1 0 0 0 0 1 1

Operation: CLR
(C)w0

CLR bit

Bytes: 2

Cycles: 1

Encoding: 1 1 0 0 0 0 1 0 bit address

Operation: CLR
(bit)w0

CPL A

Function: Complement Accumulator

Description: Each bit of the Accumulator is logically complemented (one’s complement). Bits which previ-
ously contained a one are changed to a zero and vice-versa. No flags are affected.

Example: The Accumulator contains 5CH (01011100B). The instruction,

CPL A

will leave the Accumulator set to 0A3H (10100011B).

Bytes: 1

Cycles: 1

Encoding: 1 1 1 1 0 1 0 0

Operation: CPL
(A)ws (A)

CPL bit

Function: Complement bit

Description: The bit variable specified is complemented. A bit which had been a one is changed to zero and
vice-versa. No other flags are affected. CLR can operate on the carry or any directly address-
able bit.

Note: When this instruction is used to modify an output pin, the value used as the original data
will be read from the output data latch, not the input pin.

Example: Port 1 has previously been written with 5BH (01011101B). The instruction sequence,

CPL P1.1

CPL P1.2

will leave the port set to 5BH (01011011B).

CPL C

Bytes: 1

Cycles: 1

Encoding: 1 0 1 1 0 0 1 1

Operation: CPL
(C)ws (C)

CPL bit

Bytes: 2

Cycles: 1

Encoding: 1 0 1 1 0 0 1 0 bit address

Operation: CPL
(bit)ws (bit)

DA A

Function: Decimal-adjust Accumulator for Addition

Description: DA A adjusts the eight-bit value in the Accumulator resulting from the earlier addition of two
variables (each in packed-BCD format), producing two four-bit digits. Any ADD or ADDC
instruction may have been used to perform the addition.

If Accumulator bits 3-0 are greater than nine (xxxx1010-xxxx1111), or if the AC flag is one,
six is added to the Accumulator producing the proper BCD digit in the low-order nibble. This
internal addition would set the carry flag if a carry-out of the low-order four-bit field propagat-
ed through all high-order bits, but it would not clear the carry flag otherwise.

If the carry flag is now set, or if the four high-order bits now exceed nine (1010xxxx-111xxxx),
these high-order bits are incremented by six, producing the proper BCD digit in the high-order
nibble. Again, this would set the carry flag if there was a carry-out of the high-order bits, but
wouldn’t clear the carry. The carry flag thus indicates if the sum of the original two BCD
variables is greater than 100, allowing multiple precision decimal addition. OV is not affected.

All of this occurs during the one instruction cycle. Essentially, this instruction performs the
decimal conversion by adding 00H, 06H, 60H, or 66H to the Accumulator, depending on
initial Accumulator and PSW conditions.

Note: DA A cannot simply convert a hexadecimal number in the Accumulator to BCD nota-
tion, nor does DA A apply to decimal subtraction.

37



MCSÉ-51 PROGRAMMER’S GUIDE AND INSTRUCTION SET

Example: The Accumulator holds the value 56H (01010110B) representing the packed BCD digits of the
decimal number 56. Register 3 contains the value 67H (01100111B) representing the packed
BCD digits of the decimal number 67. The carry flag is set. The instruction sequence.

ADDC A,R3
DA A

will first perform a standard twos-complement binary addition, resulting in the value 0BEH
(10111110) in the Accumulator. The carry and auxiliary carry flags will be cleared.

The Decimal Adjust instruction will then alter the Accumulator to the value 24H
(00100100B), indicating the packed BCD digits of the decimal number 24, the low-order two
digits of the decimal sum of 56, 67, and the carry-in. The carry flag will be set by the Decimal
Adjust instruction, indicating that a decimal overflow occurred. The true sum 56, 67, and 1 is
124.

BCD variables can be incremented or decremented by adding 01H or 99H. If the Accumulator
initially holds 30H (representing the digits of 30 decimal), then the instruction sequence,

ADD A,Ý99H

DA A

will leave the carry set and 29H in the Accumulator, since 30 a 99 e 129. The low-order
byte of the sum can be interpreted to mean 30 b 1 e 29.

Bytes: 1

Cycles: 1

Encoding: 1 1 0 1 0 1 0 0

Operation: DA
-contents of Accumulator are BCD
IF [[(A3-0) l 9] ¶ [(AC) e 1]]

THEN(A3-0)w(A3-0) a 6
AND

IF [[(A7-4) l 9] ¶ [(C) e 1]]
THEN (A7-4)w(A7-4) a 6

DEC byte

Function: Decrement

Description: The variable indicated is decremented by 1. An original value of 00H will underflow to 0FFH.
No flags are affected. Four operand addressing modes are allowed: accumulator, register,
direct, or register-indirect.

Note: When this instruction is used to modify an output port, the value used as the original
port data will be read from the output data latch, not the input pins.

Example: Register 0 contains 7FH (01111111B). Internal RAM locations 7EH and 7FH contain 00H
and 40H, respectively. The instruction sequence,

DEC @R0

DEC R0

DEC @R0

will leave register 0 set to 7EH and internal RAM locations 7EH and 7FH set to 0FFH and
3FH.

DEC A

Bytes: 1

Cycles: 1

Encoding: 0 0 0 1 0 1 0 0

Operation: DEC
(A)w(A) b 1

DEC Rn

Bytes: 1

Cycles: 1

Encoding: 0 0 0 1 1 r r r

Operation: DEC
(Rn)w(Rn) b 1

DEC direct

Bytes: 2

Cycles: 1

Encoding: 0 0 0 1 0 1 0 1 direct address

Operation: DEC
(direct)w(direct) b 1

DEC @Ri

Bytes: 1

Cycles: 1

Encoding: 0 0 0 1 0 1 1 i

Operation: DEC
((Ri))w((Ri)) b 1

DIV AB

Function: Divide

Description: DIV AB divides the unsigned eight-bit integer in the Accumulator by the unsigned eight-bit
integer in register B. The Accumulator receives the integer part of the quotient; register B
receives the integer remainder. The carry and OV flags will be cleared.

Exception: if B had originally contained 00H, the values returned in the Accumulator and B-
register will be undefined and the overflow flag will be set. The carry flag is cleared in any
case.

Example: The Accumulator contains 251 (0FBH or 11111011B) and B contains 18 (12H or 00010010B).
The instruction,

DIV AB

will leave 13 in the Accumulator (0DH or 00001101B) and the value 17 (11H or 00010001B)
in B, since 251 e (13 X 18) a 17. Carry and OV will both be cleared.

Bytes: 1

Cycles: 4

Encoding: 1 0 0 0 0 1 0 0

Operation: DIV
(A)15-8w(A)/(B)(B)7-0

DJNZ kbytel,krel-addrl

Function: Decrement and Jump if Not Zero

Description: DJNZ decrements the location indicated by 1, and branches to the address indicated by the
second operand if the resulting value is not zero. An original value of 00H will underflow to
0FFH. No flags are affected. The branch destination would be computed by adding the signed
relative-displacement value in the last instruction byte to the PC, after incrementing the PC to
the first byte of the following instruction.

The location decremented may be a register or directly addressed byte.

Note: When this instruction is used to modify an output port, the value used as the original
port data will be read from the output data latch, not the input pins.

Example: Internal RAM locations 40H, 50H, and 60H contain the values 01H, 70H, and 15H, respec-
tively. The instruction sequence,

DJNZ 40H,LABELÐ1
DJNZ 50H,LABELÐ2
DJNZ 60H,LABELÐ3

will cause a jump to the instruction at label LABELÐ2 with the values 00H, 6FH, and 15H in
the three RAM locations. The first jump was not taken because the result was zero.

This instruction provides a simple way of executing a program loop a given number of times,
or for adding a moderate time delay (from 2 to 512 machine cycles) with a single instruction.
The instruction sequence,

MOV R2,Ý8
TOGGLE: CPL P1.7

DJNZ R2,TOGGLE

will toggle P1.7 eight times, causing four output pulses to appear at bit 7 of output Port 1.
Each pulse will last three machine cycles; two for DJNZ and one to alter the pin.

DJNZ Rn,rel

Bytes: 2

Cycles: 2

Encoding: 1 1 0 1 1 r r r rel. address

Operation: DJNZ
(PC)w(PC) a 2
(Rn)w(Rn) b 1
IF (Rn) l 0 or (Rn) k 0

THEN
(PC)w(PC) a rel

DJNZ direct,rel

Bytes: 3

Cycles: 2

Encoding: 1 1 0 1 0 1 0 1 direct address rel. address

Operation: DJNZ
(PC)w(PC) a 2
(direct)w(direct) b 1
IF (direct) l 0 or (direct) k 0

THEN
(PC)w(PC) a rel

INC kbytel

Function: Increment

Description: INC increments the indicated variable by 1. An original value of 0FFH will overflow to 00H.
No flags are affected. Three addressing modes are allowed: register, direct, or register-indirect.

Note: When this instruction is used to modify an output port, the value used as the original
port data will be read from the output data latch, not the input pins.

Example: Register 0 contains 7EH (011111110B). Internal RAM locations 7EH and 7FH contain 0FFH
and 40H, respectively. The instruction sequence,

INC @R0
INC R0
INC @R0

will leave register 0 set to 7FH and internal RAM locations 7EH and 7FH holding (respective-
ly) 00H and 41H.

INC A

Bytes: 1

Cycles: 1

Encoding: 0 0 0 0 0 1 0 0

Operation: INC
(A)w(A) a 1

INC Rn

Bytes: 1

Cycles: 1

Encoding: 0 0 0 0 1 r r r

Operation: INC
(Rn)w(Rn) a 1

INC direct

Bytes: 2

Cycles: 1

Encoding: 0 0 0 0 0 1 0 1 direct address

Operation: INC
(direct)w(direct) a 1

INC @Ri

Bytes: 1

Cycles: 1

Encoding: 0 0 0 0 0 1 1 i

Operation: INC
((Ri))w((Ri)) a 1

INC DPTR

Function: Increment Data Pointer

Description: Increment the 16-bit data pointer by 1. A 16-bit increment (modulo 216) is performed; an
overflow of the low-order byte of the data pointer (DPL) from 0FFH to 00H will increment
the high-order byte (DPH). No flags are affected.

This is the only 16-bit register which can be incremented.

Example: Registers DPH and DPL contain 12H and 0FEH, respectively. The instruction sequence,

INC DPTR
INC DPTR
INC DPTR

will change DPH and DPL to 13H and 01H.

Bytes: 1

Cycles: 2

Encoding: 1 0 1 0 0 0 1 1

Operation: INC
(DPTR)w(DPTR) a 1

JB bit,rel

Function: Jump if Bit set

Description: If the indicated bit is a one, jump to the address indicated; otherwise proceed with the next
instruction. The branch destination is computed by adding the signed relative-displacement in
the third instruction byte to the PC, after incrementing the PC to the first byte of the next
instruction. The bit tested is not modified. No flags are affected.

Example: The data present at input port 1 is 11001010B. The Accumulator holds 56 (01010110B). The
instruction sequence,

JB P1.2,LABEL1

JB ACC.2,LABEL2

will cause program execution to branch to the instruction at label LABEL2.

Bytes: 3

Cycles: 2

Encoding: 0 0 1 0 0 0 0 0 bit address rel. address

Operation: JB
(PC)w(PC) a 3
IF (bit) e 1

THEN
(PC)w(PC) a rel

JBC bit,rel

Function: Jump if Bit is set and Clear bit

Description: If the indicated bit is one, branch to the address indicated; otherwise proceed with the next
instruction. The bit will not be cleared if it is already a zero. The branch destination is comput-
ed by adding the signed relative-displacement in the third instruction byte to the PC, after
incrementing the PC to the first byte of the next instruction. No flags are affected.

Note: When this instruction is used to test an output pin, the value used as the original data
will be read from the output data latch, not the input pin.

Example: The Accumulator holds 56H (01010110B). The instruction sequence,

JBC ACC.3,LABEL1
JBC ACC.2,LABEL2

will cause program execution to continue at the instruction identified by the label LABEL2,
with the Accumulator modified to 52H (01010010B).

44



MCSÉ-51 PROGRAMMER’S GUIDE AND INSTRUCTION SET

Bytes: 3

Cycles: 2

Encoding: 0 0 0 1 0 0 0 0 bit address rel. address

Operation: JBC
(PC)w(PC) a 3
IF (bit) e 1

THEN
(bit)w0
(PC)w(PC) a rel

JC rel

Function: Jump if Carry is set

Description: If the carry flag is set, branch to the address indicated; otherwise proceed with the next
instruction. The branch destination is computed by adding the signed relative-displacement in
the second instruction byte to the PC, after incrementing the PC twice. No flags are affected.

Example: The carry flag is cleared. The instruction sequence,

JC LABEL1
CPL C
JC LABEL 2

will set the carry and cause program execution to continue at the instruction identified by the
label LABEL2.

Bytes: 2

Cycles: 2

Encoding: 0 1 0 0 0 0 0 0 rel. address

Operation: JC
(PC)w(PC) a 2
IF (C) e 1

THEN
(PC)w(PC) a rel

JMP @AaDPTR

Function: Jump indirect

Description: Add the eight-bit unsigned contents of the Accumulator with the sixteen-bit data pointer, and
load the resulting sum to the program counter. This will be the address for subsequent instruc-
tion fetches. Sixteen-bit addition is performed (modulo 216): a carry-out from the low-order
eight bits propagates through the higher-order bits. Neither the Accumulator nor the Data
Pointer is altered. No flags are affected.

Example: An even number from 0 to 6 is in the Accumulator. The following sequence of instructions will
branch to one of four AJMP instructions in a jump table starting at JMPÐTBL:

MOV DPTR,ÝJMPÐTBL
JMP @AaDPTR

JMPÐTBL: AJMP LABEL0
AJMP LABEL1
AJMP LABEL2
AJMP LABEL3

If the Accumulator equals 04H when starting this sequence, execution will jump to label
LABEL2. Remember that AJMP is a two-byte instruction, so the jump instructions start at
every other address.

Bytes: 1

Cycles: 2

Encoding: 0 1 1 1 0 0 1 1

Operation: JMP
(PC)w(A) a (DPTR)

JNB bit,rel

Function: Jump if Bit Not set

Description: If the indicated bit is a zero, branch to the indicated address; otherwise proceed with the next
instruction. The branch destination is computed by adding the signed relative-displacement in
the third instruction byte to the PC, after incrementing the PC to the first byte of the next
instruction. The bit tested is not modified. No flags are affected.

Example: The data present at input port 1 is 11001010B. The Accumulator holds 56H (01010110B). The
instruction sequence,

JNB P1.3,LABEL1
JNB ACC.3,LABEL2

will cause program execution to continue at the instruction at label LABEL2.

Bytes: 3

Cycles: 2

Encoding: 0 0 1 1 0 0 0 0 bit address rel. address

Operation: JNB
(PC)w(PC) a 3
IF (bit) e 0

THEN (PC)w(PC) a rel.

JNC rel

Function: Jump if Carry not set

Description: If the carry flag is a zero, branch to the address indicated; otherwise proceed with the next
instruction. The branch destination is computed by adding the signed relative-displacement in
the second instruction byte to the PC, after incrementing the PC twice to point to the next
instruction. The carry flag is not modified.

Example: The carry flag is set. The instruction sequence,

JNC LABEL1
CPL C
JNC LABEL2

will clear the carry and cause program execution to continue at the instruction identified by
the label LABEL2.

Bytes: 2

Cycles: 2

Encoding: 0 1 0 1 0 0 0 0 rel. address

Operation: JNC
(PC)w(PC) a 2
IF (C) e 0

THEN (PC)w(PC) a rel

JNZ rel

Function: Jump if Accumulator Not Zero

Description: If any bit of the Accumulator is a one, branch to the indicated address; otherwise proceed with
the next instruction. The branch destination is computed by adding the signed relative-dis-
placement in the second instruction byte to the PC, after incrementing the PC twice. The
Accumulator is not modified. No flags are affected.

Example: The Accumulator originally holds 00H. The instruction sequence,

JNZ LABEL1
INC A
JNZ LABEL2

will set the Accumulator to 01H and continue at label LABEL2.

Bytes: 2

Cycles: 2

Encoding: 0 1 1 1 0 0 0 0 rel. address

Operation: JNZ
(PC)w(PC) a 2
IF (A) i 0

THEN (PC)w(PC) a rel

JZ rel

Function: Jump if Accumulator Zero

Description: If all bits of the Accumulator are zero, branch to the address indicated; otherwise proceed with
the next instruction. The branch destination is computed by adding the signed relative-dis-
placement in the second instruction byte to the PC, after incrementing the PC twice. The
Accumulator is not modified. No flags are affected.

Example: The Accumulator originally contains 01H. The instruction sequence,

JZ LABEL1
DEC A
JZ LABEL2

will change the Accumulator to 00H and cause program execution to continue at the instruc-
tion identified by the label LABEL2.

Bytes: 2

Cycles: 2

Encoding: 0 1 1 0 0 0 0 0 rel. address

Operation: JZ
(PC)w(PC) a 2
IF (A) e 0

THEN (PC)w(PC) a rel

LCALL addr16

Function: Long call

Description: LCALL calls a subroutine located at the indicated address. The instruction adds three to the
program counter to generate the address of the next instruction and then pushes the 16-bit
result onto the stack (low byte first), incrementing the Stack Pointer by two. The high-order
and low-order bytes of the PC are then loaded, respectively, with the second and third bytes of
the LCALL instruction. Program execution continues with the instruction at this address. The
subroutine may therefore begin anywhere in the full 64K-byte program memory address space.
No flags are affected.

Example: Initially the Stack Pointer equals 07H. The label ‘‘SUBRTN’’ is assigned to program memory
location 1234H. After executing the instruction,

LCALL SUBRTN

at location 0123H, the Stack Pointer will contain 09H, internal RAM locations 08H and 09H
will contain 26H and 01H, and the PC will contain 1234H.

Bytes: 3

Cycles: 2

Encoding: 0 0 0 1 0 0 1 0 addr15-addr8 addr7-addr0

Operation: LCALL
(PC)w(PC) a 3
(SP)w(SP) a 1
((SP))w(PC7-0)
(SP)w(SP) a 1
((SP))w(PC15-8)
(PC)waddr15-0

LJMP addr16

Function: Long Jump

Description: LJMP causes an unconditional branch to the indicated address, by loading the high-order and
low-order bytes of the PC (respectively) with the second and third instruction bytes. The
destination may therefore be anywhere in the full 64K program memory address space. No
flags are affected.

Example: The label ‘‘JMPADR’’ is assigned to the instruction at program memory location 1234H. The
instruction,

LJMP JMPADR

at location 0123H will load the program counter with 1234H.

Bytes: 3

Cycles: 2

Encoding: 0 0 0 0 0 0 1 0 addr15-addr8 addr7-addr0

Operation: LJMP
(PC)waddr15-0

MOV kdest-bytel,ksrc-bytel

Function: Move byte variable

Description: The byte variable indicated by the second operand is copied into the location specified by the
first operand. The source byte is not affected. No other register or flag is affected.

This is by far the most flexible operation. Fifteen combinations of source and destination
addressing modes are allowed.

Example: Internal RAM location 30H holds 40H. The value of RAM location 40H is 10H. The data
present at input port 1 is 11001010B (0CAH).

MOV R0,Ý30H ;R0 ke 30H
MOV A,@R0 ;A ke 40H
MOV R1,A ;R1 ke 40H
MOV B,@R1 ;B ke 10H
MOV @R1,P1 ;RAM (40H) ke 0CAH
MOV P2,P1 ;P2 Ý0CAH

leaves the value 30H in register 0, 40H in both the Accumulator and register 1, 10H in register
B, and 0CAH (11001010B) both in RAM location 40H and output on port 2.

MOV A,Rn

Bytes: 1

Cycles: 1

Encoding: 1 1 1 0 1 r r r

Operation: MOV
(A)w(Rn)

*MOV A,direct

Bytes: 2

Cycles: 1

Encoding: 1 1 1 0 0 1 0 1 direct address

Operation: MOV
(A)w(direct)

MOV A,ACC is not a valid instruction.

MOV A,@Ri

Bytes: 1

Cycles: 1

Encoding: 1 1 1 0 0 1 1 i

Operation: MOV
(A)w((Ri))

MOV A,Ýdata

Bytes: 2

Cycles: 1

Encoding: 0 1 1 1 0 1 0 0 immediate data

Operation: MOV
(A)wÝdata

MOV Rn,A

Bytes: 1

Cycles: 1

Encoding: 1 1 1 1 1 r r r

Operation: MOV
(Rn)w(A)

MOV Rn,direct

Bytes: 2

Cycles: 2

Encoding: 1 0 1 0 1 r r r direct addr.

Operation: MOV
(Rn)w(direct)

MOV Rn,Ýdata

Bytes: 2

Cycles: 1

Encoding: 0 1 1 1 1 r r r immediate data

Operation: MOV
(Rn)wÝdata

MOV direct,A

Bytes: 2

Cycles: 1

Encoding: 1 1 1 1 0 1 0 1 direct address

Operation: MOV
(direct)w(A)

MOV direct,Rn

Bytes: 2

Cycles: 2

Encoding: 1 0 0 0 1 r r r direct address

Operation: MOV
(direct)w(Rn)

MOV direct,direct

Bytes: 3

Cycles: 2

Encoding: 1 0 0 0 0 1 0 1 dir. addr. (src) dir. addr. (dest)

Operation: MOV
(direct)w(direct)

MOV direct,@Ri

Bytes: 2

Cycles: 2

Encoding: 1 0 0 0 0 1 1 i direct addr.

Operation: MOV
(direct)w((Ri))

MOV direct,Ýdata

Bytes: 3

Cycles: 2

Encoding: 0 1 1 1 0 1 0 1 direct address immediate data

Operation: MOV
(direct)wÝdata

MOV @Ri,A

Bytes: 1

Cycles: 1

Encoding: 1 1 1 1 0 1 1 i

Operation: MOV
((Ri))w(A)

MOV @Ri,direct

Bytes: 2

Cycles: 2

Encoding: 1 0 1 0 0 1 1 i direct addr.

Operation: MOV
((Ri))w(direct)

MOV @Ri,Ýdata

Bytes: 2

Cycles: 1

Encoding: 0 1 1 1 0 1 1 i immediate data

Operation: MOV
((RI))wÝdata

MOV kdest-bitl,ksrc-bitl

Function: Move bit data

Description: The Boolean variable indicated by the second operand is copied into the location specified by
the first operand. One of the operands must be the carry flag; the other may be any directly
addressable bit. No other register or flag is affected.

Example: The carry flag is originally set. The data present at input Port 3 is 11000101B. The data
previously written to output Port 1 is 35H (00110101B).

MOV P1.3,C
MOV C,P3.3
MOV P1.2,C

will leave the carry cleared and change Port 1 to 39H (00111001B).

MOV C,bit

Bytes: 2

Cycles: 1

Encoding: 1 0 1 0 0 0 1 0 bit address

Operation: MOV
(C)w(bit)

MOV bit,C

Bytes: 2

Cycles: 2

Encoding: 1 0 0 1 0 0 1 0 bit address

Operation: MOV
(bit)w(C)

MOV DPTR,Ýdata16

Function: Load Data Pointer with a 16-bit constant

Description: The Data Pointer is loaded with the 16-bit constant indicated. The 16-bit constant is loaded
into the second and third bytes of the instruction. The second byte (DPH) is the high-order
byte, while the third byte (DPL) holds the low-order byte. No flags are affected.

This is the only instruction which moves 16 bits of data at once.

Example: The instruction,

MOV DPTR,Ý1234H

will load the value 1234H into the Data Pointer: DPH will hold 12H and DPL will hold 34H.

Bytes: 3

Cycles: 2

Encoding: 1 0 0 1 0 0 0 0 immed. data15-8 immed. data7-0

Operation: MOV
(DPTR)wÝdata15-0
DPH V DPLwÝdata15-8 V Ýdata7-0

MOVC A,@Aakbase-regl

Function: Move Code byte

Description: The MOVC instructions load the Accumulator with a code byte, or constant from program
memory. The address of the byte fetched is the sum of the original unsigned eight-bit Accumu-
lator contents and the contents of a sixteen-bit base register, which may be either the Data
Pointer or the PC. In the latter case, the PC is incremented to the address of the following
instruction before being added with the Accumulator; otherwise the base register is not al-
tered. Sixteen-bit addition is performed so a carry-out from the low-order eight bits may
propagate through higher-order bits. No flags are affected.

Example: A value between 0 and 3 is in the Accumulator. The following instructions will translate the
value in the Accumulator to one of four values defined by the DB (define byte) directive.

RELÐPC: INC A

MOVC A,@AaPC

RET

DB 66H

DB 77H

DB 88H

DB 99H

If the subroutine is called with the Accumulator equal to 01H, it will return with 77H in the
Accumulator. The INC A before the MOVC instruction is needed to ‘‘get around’’ the RET
instruction above the table. If several bytes of code separated the MOVC from the table, the
corresponding number would be added to the Accumulator instead.

MOVC A,@AaDPTR

Bytes: 1

Cycles: 2

Encoding: 1 0 0 1 0 0 1 1

Operation: MOVC
(A)w((A) a (DPTR))

MOVC A,@A a PC

Bytes: 1

Cycles: 2

Encoding: 1 0 0 0 0 0 1 1

Operation: MOVC
(PC)w(PC) a 1
(A)w((A) a (PC))

MOVX kdest-bytel,ksrc-bytel

Function: Move External

Description: The MOVX instructions transfer data between the Accumulator and a byte of external data
memory, hence the ‘‘X’’ appended to MOV. There are two types of instructions, differing in
whether they provide an eight-bit or sixteen-bit indirect address to the external data RAM.

In the first type, the contents of R0 or R1 in the current register bank provide an eight-bit
address multiplexed with data on P0. Eight bits are sufficient for external I/O expansion
decoding or for a relatively small RAM array. For somewhat larger arrays, any output port
pins can be used to output higher-order address bits. These pins would be controlled by an
output instruction preceding the MOVX.

In the second type of MOVX instruction, the Data Pointer generates a sixteen-bit address. P2
outputs the high-order eight address bits (the contents of DPH) while P0 multiplexes the low-
order eight bits (DPL) with data. The P2 Special Function Register retains its previous con-
tents while the P2 output buffers are emitting the contents of DPH. This form is faster and
more efficient when accessing very large data arrays (up to 64K bytes), since no additional
instructions are needed to set up the output ports.

It is possible in some situations to mix the two MOVX types. A large RAM array with its
high-order address lines driven by P2 can be addressed via the Data Pointer, or with code to
output high-order address bits to P2 followed by a MOVX instruction using R0 or R1.

Example: An external 256 byte RAM using multiplexed address/data lines (e.g., an Intel 8155 RAM/
I/O/Timer) is connected to the 8051 Port 0. Port 3 provides control lines for the external
RAM. Ports 1 and 2 are used for normal I/O. Registers 0 and 1 contain 12H and 34H.
Location 34H of the external RAM holds the value 56H. The instruction sequence,

MOVX A,@R1

MOVX @R0,A

copies the value 56H into both the Accumulator and external RAM location 12H.

MOVX A,@Ri

Bytes: 1

Cycles: 2

Encoding: 1 1 1 0 0 0 1 i

Operation: MOVX
(A)w((Ri))

MOVX A,@DPTR

Bytes: 1

Cycles: 2

Encoding: 1 1 1 0 0 0 0 0

Operation: MOVX
(A)w((DPTR))

MOVX @Ri,A

Bytes: 1

Cycles: 2

Encoding: 1 1 1 1 0 0 1 i

Operation: MOVX
((Ri))w(A)

MOVX @DPTR,A

Bytes: 1

Cycles: 2

Encoding: 1 1 1 1 0 0 0 0

Operation: MOVX
(DPTR)w(A)

MUL AB

Function: Multiply

Description: MUL AB multiplies the unsigned eight-bit integers in the Accumulator and register B. The
low-order byte of the sixteen-bit product is left in the Accumulator, and the high-order byte in
B. If the product is greater than 255 (0FFH) the overflow flag is set; otherwise it is cleared.
The carry flag is always cleared.

Example: Originally the Accumulator holds the value 80 (50H). Register B holds the value 160 (0A0H).
The instruction,

MUL AB

will give the product 12,800 (3200H), so B is changed to 32H (00110010B) and the Accumula-
tor is cleared. The overflow flag is set, carry is cleared.

Bytes: 1

Cycles: 4

Encoding: 1 0 1 0 0 1 0 0

Operation: MUL
(A)7-0w(A) X (B)
(B)15-8

NOP

Function: No Operation

Description: Execution continues at the following instruction. Other than the PC, no registers or flags are
affected.

Example: It is desired to produce a low-going output pulse on bit 7 of Port 2 lasting exactly 5 cycles. A
simple SETB/CLR sequence would generate a one-cycle pulse, so four additional cycles must
be inserted. This may be done (assuming no interrupts are enabled) with the instruction
sequence,

CLR P2.7

NOP

NOP

NOP

NOP

SETB P2.7

Bytes: 1

Cycles: 1

Encoding: 0 0 0 0 0 0 0 0

Operation: NOP
(PC)w(PC) a 1

ORL kdest-bytel ksrc-bytel

Function: Logical-OR for byte variables

Description: ORL performs the bitwise logical-OR operation between the indicated variables, storing the
results in the destination byte. No flags are affected.

The two operands allow six addressing mode combinations. When the destination is the Accu-
mulator, the source can use register, direct, register-indirect, or immediate addressing; when
the destination is a direct address, the source can be the Accumulator or immediate data.

Note: When this instruction is used to modify an output port, the value used as the original
port data will be read from the output data latch, not the input pins.

Example: If the Accumulator holds 0C3H (11000011B) and R0 holds 55H (01010101B) then the in-
struction,

ORL A,R0

will leave the Accumulator holding the value 0D7H (11010111B).

When the destination is a directly addressed byte, the instruction can set combinations of bits
in any RAM location or hardware register. The pattern of bits to be set is determined by a
mask byte, which may be either a constant data value in the instruction or a variable computed
in the Accumulator at run-time. The instruction,

ORL P1,Ý00110010B

will set bits 5, 4, and 1 of output Port 1.

ORL A,Rn

Bytes: 1

Cycles: 1

Encoding: 0 1 0 0 1 r r r

Operation: ORL
(A)w(A) ¶ (Rn)

ORL A,direct

Bytes: 2

Cycles: 1

Encoding: 0 1 0 0 0 1 0 1 direct address

Operation: ORL
(A)w(A) ¶ (direct)

ORL A,@Ri

Bytes: 1

Cycles: 1

Encoding: 0 1 0 0 0 1 1 i

Operation: ORL
(A)w(A) ¶ ((Ri))

ORL A,Ýdata

Bytes: 2

Cycles: 1

Encoding: 0 1 0 0 0 1 0 0 immediate data

Operation: ORL
(A)w(A) ¶ Ýdata

ORL direct,A

Bytes: 2

Cycles: 1

Encoding: 0 1 0 0 0 0 1 0 direct address

Operation: ORL
(direct)w(direct) ¶ (A)

ORL direct,Ýdata

Bytes: 3

Cycles: 2

Encoding: 0 1 0 0 0 0 1 1 direct addr. immediate data

Operation: ORL
(direct)w(direct) ¶ Ýdata

ORL C,ksrc-bitl

Function: Logical-OR for bit variables

Description: Set the carry flag if the Boolean value is a logical 1; leave the carry in its current state
otherwise . A slash (‘‘/’’) preceding the operand in the assembly language indicates that the
logical complement of the addressed bit is used as the source value, but the source bit itself is
not affected. No other flags are affected.

Example: Set the carry flag if and only if P1.0 e 1, ACC. 7 e 1, or OV e 0:

MOV C,P1.0 ;LOAD CARRY WITH INPUT PIN P10

ORL C,ACC.7 ;OR CARRY WITH THE ACC. BIT 7

ORL C,/OV ;OR CARRY WITH THE INVERSE OF OV.

ORL C,bit

Bytes: 2

Cycles: 2

Encoding: 0 1 1 1 0 0 1 0 bit address

Operation: ORL
(C)w(C) ¶ (bit)

ORL C,/bit

Bytes: 2

Cycles: 2

Encoding: 1 0 1 0 0 0 0 0 bit address

Operation: ORL
(C)w(C) ¶ (bit)

POP direct

Function: Pop from stack.

Description: The contents of the internal RAM location addressed by the Stack Pointer is read, and the
Stack Pointer is decremented by one. The value read is then transferred to the directly ad-
dressed byte indicated. No flags are affected.

Example: The Stack Pointer originally contains the value 32H, and internal RAM locations 30H
through 32H contain the values 20H, 23H, and 01H, respectively. The instruction sequence,

POP DPH

POP DPL

will leave the Stack Pointer equal to the value 30H and the Data Pointer set to 0123H. At this
point the instruction,

POP SP

will leave the Stack Pointer set to 20H. Note that in this special case the Stack Pointer was
decremented to 2FH before being loaded with the value popped (20H).

Bytes: 2

Cycles: 2

Encoding: 1 1 0 1 0 0 0 0 direct address

Operation: POP
(direct)w((SP))
(SP)w(SP) b 1

PUSH direct

Function: Push onto stack

Description: The Stack Pointer is incremented by one. The contents of the indicated variable is then copied
into the internal RAM location addressed by the Stack Pointer. Otherwise no flags are affect-
ed.

Example: On entering an interrupt routine the Stack Pointer contains 09H. The Data Pointer holds the
value 0123H. The instruction sequence,

PUSH DPL

PUSH DPH

will leave the Stack Pointer set to 0BH and store 23H and 01H in internal RAM locations
0AH and 0BH, respectively.

Bytes: 2

Cycles: 2

Encoding: 1 1 0 0 0 0 0 0 direct address

Operation: PUSH
(SP)w(SP) a 1
((SP))w(direct)

RET

Function: Return from subroutine

Description: RET pops the high- and low-order bytes of the PC successively from the stack, decrementing
the Stack Pointer by two. Program execution continues at the resulting address, generally the
instruction immediately following an ACALL or LCALL. No flags are affected.

Example: The Stack Pointer originally contains the value 0BH. Internal RAM locations 0AH and 0BH
contain the values 23H and 01H, respectively. The instruction,

RET

will leave the Stack Pointer equal to the value 09H. Program execution will continue at
location 0123H.

Bytes: 1

Cycles: 2

Encoding: 0 0 1 0 0 0 1 0

Operation: RET
(PC15-8)w((SP))
(SP)w(SP) b 1
(PC7-0)w((SP))
(SP)w(SP) b 1

RETI

Function: Return from interrupt

Description: RETI pops the high- and low-order bytes of the PC successively from the stack, and restores
the interrupt logic to accept additional interrupts at the same priority level as the one just
processed. The Stack Pointer is left decremented by two. No other registers are affected; the
PSW is not automatically restored to its pre-interrupt status. Program execution continues at
the resulting address, which is generally the instruction immediately after the point at which
the interrupt request was detected. If a lower- or same-level interrupt had been pending when
the RETI instruction is executed, that one instruction will be executed before the pending
interrupt is processed.

Example: The Stack Pointer originally contains the value 0BH. An interrupt was detected during the
instruction ending at location 0122H. Internal RAM locations 0AH and 0BH contain the
values 23H and 01H, respectively. The instruction,

RETI

will leave the Stack Pointer equal to 09H and return program execution to location 0123H.

Bytes: 1

Cycles: 2

Encoding: 0 0 1 1 0 0 1 0

Operation: RETI
(PC15-8)w((SP))
(SP)w(SP) b 1
(PC7-0)w((SP))
(SP)w(SP) b 1

RL A

Function: Rotate Accumulator Left

Description: The eight bits in the Accumulator are rotated one bit to the left. Bit 7 is rotated into the bit 0
position. No flags are affected.

Example: The Accumulator holds the value 0C5H (11000101B). The instruction,

RL A

leaves the Accumulator holding the value 8BH (10001011B) with the carry unaffected.

Bytes: 1

Cycles: 1

Encoding: 0 0 1 0 0 0 1 1

Operation: RL
(An a 1)w(An) n e 0 b 6
(A0)w(A7)

RLC A

Function: Rotate Accumulator Left through the Carry flag

Description: The eight bits in the Accumulator and the carry flag are together rotated one bit to the left. Bit
7 moves into the carry flag; the original state of the carry flag moves into the bit 0 position. No
other flags are affected.

Example: The Accumulator holds the value 0C5H (11000101B), and the carry is zero. The instruction,

RLC A

leaves the Accumulator holding the value 8BH (10001010B) with the carry set.

Bytes: 1

Cycles: 1

Encoding: 0 0 1 1 0 0 1 1

Operation: RLC
(An a 1)w(An) n e 0 b 6
(A0)w(C)
(C)w(A7)

RR A

Function: Rotate Accumulator Right

Description: The eight bits in the Accumulator are rotated one bit to the right. Bit 0 is rotated into the bit 7
position. No flags are affected.

Example: The Accumulator holds the value 0C5H (11000101B). The instruction,

RR A

leaves the Accumulator holding the value 0E2H (11100010B) with the carry unaffected.

Bytes: 1

Cycles: 1

Encoding: 0 0 0 0 0 0 1 1

Operation: RR
(An)w(An a 1) n e 0 b 6
(A7)w(A0)

RRC A

Function: Rotate Accumulator Right through Carry flag

Description: The eight bits in the Accumulator and the carry flag are together rotated one bit to the right.
Bit 0 moves into the carry flag; the original value of the carry flag moves into the bit 7
position. No other flags are affected.

Example: The Accumulator holds the value 0C5H (11000101B), the carry is zero. The instruction,

RRC A

leaves the Accumulator holding the value 62 (01100010B) with the carry set.

Bytes: 1

Cycles: 1

Encoding: 0 0 0 1 0 0 1 1

Operation: RRC
(An)w(An a 1) n e 0 b 6
(A7)w(C)
(C)w(A0)

SETB kbitl

Function: Set Bit

Description: SETB sets the indicated bit to one. SETB can operate on the carry flag or any directly
addressable bit. No other flags are affected.

Example: The carry flag is cleared. Output Port 1 has been written with the value 34H (00110100B). The
instructions,

SETB C

SETB P1.0

will leave the carry flag set to 1 and change the data output on Port 1 to 35H (00110101B).

SETB C

Bytes: 1

Cycles: 1

Encoding: 1 1 0 1 0 0 1 1

Operation: SETB
(C)w1

SETB bit

Bytes: 2

Cycles: 1

Encoding: 1 1 0 1 0 0 1 0 bit address

Operation: SETB
(bit)w1

SJMP rel

Function: Short Jump

Description: Program control branches unconditionally to the address indicated. The branch destination is
computed by adding the signed displacement in the second instruction byte to the PC, after
incrementing the PC twice. Therefore, the range of destinations allowed is from 128 bytes
preceding this instruction to 127 bytes following it.

Example: The label ‘‘RELADR’’ is assigned to an instruction at program memory location 0123H. The
instruction,

SJMP RELADR

will assemble into location 0100H. After the instruction is executed, the PC will contain the
value 0123H.

(Note: Under the above conditions the instruction following SJMP will be at 102H. Therefore,
the displacement byte of the instruction will be the relative offset (0123H-0102H) e 21H. Put
another way, an SJMP with a displacement of 0FEH would be a one-instruction infinite loop.)

Bytes: 2

Cycles: 2

Encoding: 1 0 0 0 0 0 0 0 rel. address

Operation: SJMP
(PC)w(PC) a 2
(PC)w(PC) a rel

SUBB A,ksrc-bytel

Function: Subtract with borrow

Description: SUBB subtracts the indicated variable and the carry flag together from the Accumulator,
leaving the result in the Accumulator. SUBB sets the carry (borrow) flag if a borrow is needed
for bit 7, and clears C otherwise. (If C was set before executing a SUBB instruction, this
indicates that a borrow was needed for the previous step in a multiple precision subtraction, so
the carry is subtracted from the Accumulator along with the source operand.) AC is set if a
borrow is needed for bit 3, and cleared otherwise. OV is set if a borrow is needed into bit 6, but
not into bit 7, or into bit 7, but not bit 6.

When subtracting signed integers OV indicates a negative number produced when a negative
value is subtracted from a positive value, or a positive result when a positive number is
subtracted from a negative number.

The source operand allows four addressing modes: register, direct, register-indirect, or imme-
diate.

Example: The Accumulator holds 0C9H (11001001B), register 2 holds 54H (01010100B), and the carry
flag is set. The instruction,

SUBB A,R2

will leave the value 74H (01110100B) in the accumulator, with the carry flag and AC cleared
but OV set.

Notice that 0C9H minus 54H is 75H. The difference between this and the above result is due
to the carry (borrow) flag being set before the operation. If the state of the carry is not known
before starting a single or multiple-precision subtraction, it should be explicitly cleared by a
CLR C instruction.

SUBB A,Rn

Bytes: 1

Cycles: 1

Encoding: 1 0 0 1 1 r r r

Operation: SUBB
(A)w(A) b (C) b (Rn)

SUBB A,direct

Bytes: 2

Cycles: 1

Encoding: 1 0 0 1 0 1 0 1 direct address

Operation: SUBB
(A)w(A) b (C) b (direct)

SUBB A,@Ri

Bytes: 1

Cycles: 1

Encoding: 1 0 0 1 0 1 1 i

Operation: SUBB
(A)w(A) b (C) b ((Ri))

SUBB A,Ýdata

Bytes: 2

Cycles: 1

Encoding: 1 0 0 1 0 1 0 0 immediate data

Operation: SUBB
(A)w(A) b (C) b Ýdata

SWAP A

Function: Swap nibbles within the Accumulator

Description: SWAP A interchanges the low- and high-order nibbles (four-bit fields) of the Accumulator
(bits 3-0 and bits 7-4). The operation can also be thought of as a four-bit rotate instruction. No
flags are affected.

Example: The Accumulator holds the value 0C5H (11000101B). The instruction,

SWAP A

leaves the Accumulator holding the value 5CH (01011100B).

Bytes: 1

Cycles: 1

Encoding: 1 1 0 0 0 1 0 0

Operation: SWAP
(A3-0)

x
w (A7-4)

XCH A,kbytel

Function: Exchange Accumulator with byte variable

Description: XCH loads the Accumulator with the contents of the indicated variable, at the same time
writing the original Accumulator contents to the indicated variable. The source/destination
operand can use register, direct, or register-indirect addressing.

Example: R0 contains the address 20H. The Accumulator holds the value 3FH (00111111B). Internal
RAM location 20H holds the value 75H (01110101B). The instruction,

XCH A,@R0

will leave RAM location 20H holding the values 3FH (00111111B) and 75H (01110101B) in
the accumulator.

XCH A,Rn

Bytes: 1

Cycles: 1

Encoding: 1 1 0 0 1 r r r

Operation: XCH
(A) x

w (Rn)

XCH A,direct

Bytes: 2

Cycles: 1

Encoding: 1 1 0 0 0 1 0 1 direct address

Operation: XCH
(A) x

w (direct)

XCH A,@Ri

Bytes: 1

Cycles: 1

Encoding: 1 1 0 0 0 1 1 i

Operation: XCH
(A) x

w ((Ri))

XCHD A,@Ri

Function: Exchange Digit

Description: XCHD exchanges the low-order nibble of the Accumulator (bits 3-0), generally representing a
hexadecimal or BCD digit, with that of the internal RAM location indirectly addressed by the
specified register. The high-order nibbles (bits 7-4) of each register are not affected. No flags
are affected.

Example: R0 contains the address 20H. The Accumulator holds the value 36H (00110110B). Internal
RAM location 20H holds the value 75H (01110101B). The instruction,

XCHD A,@R0

will leave RAM location 20H holding the value 76H (01110110B) and 35H (00110101B) in the
Accumulator.

Bytes: 1

Cycles: 1

Encoding: 1 1 0 1 0 1 1 i

Operation: XCHD
(A3-0)

x
w ((Ri3-0))

XRL kdest-bytel,ksrc-bytel

Function: Logical Exclusive-OR for byte variables

Description: XRL performs the bitwise logical Exclusive-OR operation between the indicated variables,
storing the results in the destination. No flags are affected.

The two operands allow six addressing mode combinations. When the destination is the Accu-
mulator, the source can use register, direct, register-indirect, or immediate addressing; when
the destination is a direct address, the source can be the Accumulator or immediate data.

(Note: When this instruction is used to modify an output port, the value used as the original
port data will be read from the output data latch, not the input pins.)

Example: If the Accumulator holds 0C3H (11000011B) and register 0 holds 0AAH (10101010B) then
the instruction,

XRL A,R0

will leave the Accumulator holding the value 69H (01101001B).

When the destination is a directly addressed byte, this instruction can complement combina-
tions of bits in any RAM location or hardware register. The pattern of bits to be complement-
ed is then determined by a mask byte, either a constant contained in the instruction or a
variable computed in the Accumulator at run-time. The instruction,

XRL P1,Ý00110001B

will complement bits 5, 4, and 0 of output Port 1.

XRL A,Rn

Bytes: 1

Cycles: 1

Encoding: 0 1 1 0 1 r r r

Operation: XRL
(A)w(A) I (Rn)

XRL A,direct

Bytes: 2

Cycles: 1

Encoding: 0 1 1 0 0 1 0 1 direct address

Operation: XRL
(A)w(A) I (direct)

XRL A,@Ri

Bytes: 1

Cycles: 1

Encoding: 0 1 1 0 0 1 1 i

Operation: XRL
(A)w(A) I ((Ri))

XRL A,Ýdata

Bytes: 2

Cycles: 1

Encoding: 0 1 1 0 0 1 0 0 immediate data

Operation: XRL
(A)w(A) I Ýdata

XRL direct,A

Bytes: 2

Cycles: 1

Encoding: 0 1 1 0 0 0 1 0 direct address

Operation: XRL
(direct)w(direct) I (A)

XRL direct,Ýdata

Bytes: 3

Cycles: 2

Encoding: 0 1 1 0 0 0 1 1 direct address immediate data

Operation: XRL
(direct)w(direct) I Ýdata

